﻿// Upgrade NOTE: replaced '_Object2World' with 'unity_ObjectToWorld'
// Upgrade NOTE: replaced '_World2Object' with 'unity_WorldToObject'
// Upgrade NOTE: replaced 'mul(UNITY_MATRIX_MVP,*)' with 'UnityObjectToClipPos(*)'

Shader "Custom/SpecularPixelLevel"
{
Properties
	{
		_Diffuse ("Diffuse" , Color)=(1, 1, 1, 1)
		_Specular ("Specular", Color)=(1, 1, 1, 1)
		_Gloss ("Gloss",Range(8.0 ,256.0))= 20
	}
 
	SubShader
	{
		Pass
		{
			Tags{"LightMode"="ForwardBase"}
			
			CGPROGRAM
 
			#pragma vertex vert
			#pragma fragment frag
 
			#include "Lighting.cginc"
 
			fixed4 _Diffuse;
			fixed4 _Specular;
			float _Gloss;
 
			struct a2v
			{
				float4 vertex : POSITION;
				float4 normal : NORMAL;
			};
 
			struct v2f
			{
				float4 pos : SV_POSITION;
				float3 worldNormal : TEXCOORD0;
				float3 worldPos : TEXCOORD1;
			};
 
			v2f vert (a2v v)
			{
				v2f o;
 
				// Transform the vertex from object space to projection space
				//把局部顶点光照投射到空间
				o.pos = UnityObjectToClipPos(v.vertex);
 
				//Transform the normal from object  space to world  space;
				//把局部法线转换到空间法线
				o.worldNormal = mul(v.normal, (float3x3)unity_WorldToObject);
 
				//Transform the vertex from obeject space to world space;
				//把局部顶点光照转换到空间中
				o.worldPos = mul(unity_ObjectToWorld , v.vertex).xyz;
 
				return o;
			}
 
			fixed4 frag (v2f i) : SV_Target
			{
				//Get ambient term
				fixed3 ambient = UNITY_LIGHTMODEL_AMBIENT.xyz;
 
				fixed3 worldNormal = normalize (i.worldNormal);
 
				fixed3 worldLightDir = normalize (_WorldSpaceLightPos0.xyz);
				
				//Coumpute diffuse term				
				fixed3 diffuse = _LightColor0.rgb * _Diffuse .rgb * saturate(dot(worldNormal,worldLightDir));
 
 
				//Get the reflect direction in world space;
				//获取空间中的光照方向
				fixed3 reflectDir = normalize (reflect (-worldLightDir,worldNormal));
 
				//Get view direction in world space;
				//获得视窗的光照方向
				fixed3 viewDir = normalize(_WorldSpaceCameraPos.xyz - i.worldPos.xyz);
 
				//Compute sepcular term
				fixed3 specular = _LightColor0.rgb * _Specular.rgb * pow(saturate(dot(reflectDir,viewDir)),_Gloss);
 
				return fixed4(ambient + diffuse + specular, 1.0);
			}
			ENDCG
		}
	}
	FallBack "Specular"

}
